Pensieve header: Computing the Jones polynomial, rough procedure.

First, Class Introduction as in http://drorbn.net/AcademicPensieve/Classes/17-1350-AKT/About.html.

The Jones Polynomial of the Trefoil Knot



```
<< KnotTheory`
```

```
Loading KnotTheory` version of September 6, 2014, 13:37:37.2841.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
K = Knot[4, 1]
```

```
Knot[4, 1]
```

```
PD[K]
```

```
PD[X[4, 2, 5, 1], X[8, 6, 1, 5], X[6, 3, 7, 4], X[2, 7, 3, 8]]
```

```
PositiveQ[X[3, 6, 4, 1]]
```

```
False
```

```
(x + y) ^2
```

$(x+y)^2$

```
(x + y) ^2 // TreeForm
```




```
(x + y) ^2 /. y → z
```

$(x+z)^2$

```
(x + y) ^2 /. 2 → 3.14
```

$(x+y)^{3.14}$

```
{1, 2, 3, 4} /. _ :→ 7
```

7

```
{1, 2, 3, 4} /. _Integer :→ 7
```

{7, 7, 7, 7}

```
{1, 2, 3, 4} /. Travis_Integer :→ 7
```

{7, 7, 7, 7}

```
{1, 2, 3, 4} /. Travis_Integer :→ Travis^3
```

{1, 8, 27, 64}

```
f @@ ((x + y) ^2)
```

f[x + y, 2]

```
(x + y) ^2
```

$(x+y)^2$

```
Power[Plus[x, y], 2]
```

$(x+y)^2$

```
PD[K]
```

PD[X[4, 2, 5, 1], X[8, 6, 1, 5], X[6, 3, 7, 4], X[2, 7, 3, 8]]

```
t1 = PD[K] /. x_X :> If[PositiveQ[x], Xp @@ x, Xm @@ x]
```

PD[Xp[4, 2, 5, 1], Xp[8, 6, 1, 5], Xm[6, 3, 7, 4], Xm[2, 7, 3, 8]]

```
t1
```

PD[Xp[4, 2, 5, 1], Xp[8, 6, 1, 5], Xm[6, 3, 7, 4], Xm[2, 7, 3, 8]]

```
t2 = Times @@ t1
```

Xm[2, 7, 3, 8] Xm[6, 3, 7, 4] Xp[4, 2, 5, 1] Xp[8, 6, 1, 5]

```
t3 = t2 /. {
   Xp[i_, j_, k_, l_] :> q * P[i, j] P[k, l] - q^2 * P[i, l] P[j, k],
   Xm[i_, j_, k_, l_] :> -q^(-2) * P[i, j] P[k, l] + 1/q * P[i, l] P[j, k]
  }
```

$$\left(q\,P[1,5]\,P[2,4]-q^2\,P[1,4]\,P[2,5]\right)\left(\frac{P[2,8]\,P[3,7]}{q}-\frac{P[2,7]\,P[3,8]}{q^2}\right)\left(\frac{P[3,7]\,P[4,6]}{q}-\frac{P[3,6]\,P[4,7]}{q^2}\right)\left(-q^2\,P[1,6]\,P[5,8]+q\,P[1,5]\,P[6,8]\right)$$

```
Expand[(a + b) (c + d)]
```

$a\,c+b\,c+a\,d+b\,d$

```
Expand[(x + y)^10]
```

$$x^{10}+10\,x^9\,y+45\,x^8\,y^2+120\,x^7\,y^3+210\,x^6\,y^4+252\,x^5\,y^5+210\,x^4\,y^6+120\,x^3\,y^7+45\,x^2\,y^8+10\,x\,y^9+y^{10}$$

```
t4 = Expand[t3]
```

$-q P[1,5] P[1,6] P[2,4] P[2,8] P[3,7]^2 P[4,6] P[5,8] + q^2 P[1,4] P[1,6] P[2,5] P[2,8] P[3,7]^2 P[4,6] P[5,8] + P[1,5] P[1,6] P[2,4] P[2,7] P[3,7] P[3,8] P[4,6] P[5,8] - q P[1,4] P[1,6] P[2,5] P[2,7] P[3,7] P[3,8] P[4,6] P[5,8] + P[1,5] P[1,6] P[2,4] P[2,8] P[3,6] P[3,7] P[4,7] P[5,8] - q P[1,4] P[1,6] P[2,5] P[2,8] P[3,6] P[3,7] P[4,7] P[5,8] - \frac{1}{q} P[1,5] P[1,6] P[2,4] P[2,7] P[3,6] P[3,8] P[4,7] P[5,8] + P[1,4] P[1,6] P[2,5] P[2,7] P[3,6] P[3,8] P[4,7] P[5,8] + P[1,5]^2 P[2,4] P[2,8] P[3,7]^2 P[4,6] P[6,8] - q P[1,4] P[1,5] P[2,5] P[2,8] P[3,7]^2 P[4,6] P[6,8] - \frac{1}{q} P[1,5]^2 P[2,4] P[2,7] P[3,7] P[3,8] P[4,6] P[6,8] + P[1,4] P[1,5] P[2,5] P[2,7] P[3,7] P[3,8] P[4,6] P[6,8] - \frac{1}{q} P[1,5]^2 P[2,4] P[2,8] P[3,6] P[3,7] P[4,7] P[6,8] + P[1,4] P[1,5] P[2,5] P[2,8] P[3,6] P[3,7] P[4,7] P[6,8] + \frac{1}{q^2} P[1,5]^2 P[2,4] P[2,7] P[3,6] P[3,8] P[4,7] P[6,8] - \frac{1}{q} P[1,4] P[1,5] P[2,5] P[2,7] P[3,6] P[3,8] P[4,7] P[6,8]$

```
{PP[1, 3], PP[3, 1]}
```

{PP[1, 3], PP[3, 1]}

```
SetAttributes[PP, Orderless]
```

```
{PP[1, 3], PP[3, 1]}
```

{PP[1, 3], PP[1, 3]}

```
SetAttributes[P, Orderless]
```

```
t5 = t4 //. P[a_, b_] P[b_, c_] :> P[a, c]
```

$q^2 P[3,7]^2 P[4,6]^2 P[5,8]^2 - q P[5,8]^2 P[6,7]^2 - 2 q P[3,7]^2 P[6,8]^2 + P[1,5]^2 P[3,7]^2 P[6,8]^2 - \frac{P[4,7]^2 P[6,8]^2}{q} + \frac{1}{q^2} P[1,5]^2 P[4,7]^2 P[6,8]^2 - \frac{P[5,7]^2 P[6,8]^2}{q} + 5 P[7,8]^2 - \frac{2 P[1,5]^2 P[7,8]^2}{q} - q P[4,6]^2 P[7,8]^2$

```
t6 = t5 /. {
    P[i_, i_] :> (q + 1/q),
    P[i_, j_]^2 :> (q + 1/q)
  }
```

$5 \left(\frac{1}{q} + q\right) - \frac{4 \left(\frac{1}{q} + q\right)^2}{q} - 4 q \left(\frac{1}{q} + q\right)^2 + \left(\frac{1}{q} + q\right)^3 + \frac{\left(\frac{1}{q} + q\right)^3}{q^2} + q^2 \left(\frac{1}{q} + q\right)^3$

```
Expand[t6]
```

$\frac{1}{q^5} + q^5$

```
Simplify[t6]
```

$\frac{1}{q^5} + q^5$

```
Jones[Knot[4, 1]][q]
```

KnotTheory: Loading precomputed data in Jones4Knots`.

$1 + \frac{1}{q^2} - \frac{1}{q} - q + q^2$

```
Simplify[t6/(q + q^-1) /. q → √q]
```

$1 + \frac{1}{q^2} - \frac{1}{q} - q + q^2$

A Jones Polynomial Program

```
J[K_] := Module[{t1, t2, t3, t4, t5, t6, P},
  SetAttributes[P, Orderless];
  t1 = PD[K] /. x_X :> If[PositiveQ[x], Xp @@ x, Xm @@ x];
  t2 = Times @@ t1;
  t3 = t2 /. {
     Xp[i_, j_, k_, l_] :> q * P[i, j] P[k, l] - q^2 * P[i, l] P[j, k],
     Xm[i_, j_, k_, l_] :> -q^(-2) * P[i, j] P[k, l] + 1/q * P[i, l] P[j, k]
    };
  t4 = Expand[t3];
  t5 = t4 //. P[a_, b_] P[b_, c_] :> P[a, c];
  t6 = t5 /. {
     P[i_, i_] :> (q + 1/q),
     P[i_, j_]^2 :> (q + 1/q)
    };
  Simplify[t6/(q + q^-1) /. q → √q]
 ]
```

```
J[Knot[8, 17]]
```

$7 + \frac{1}{q^4} - \frac{3}{q^3} + \frac{5}{q^2} - \frac{6}{q} - 6\,q + 5\,q^2 - 3\,q^3 + q^4$

```
Jones[Knot[8, 17]][q]
```

$7 + \frac{1}{q^4} - \frac{3}{q^3} + \frac{5}{q^2} - \frac{6}{q} - 6\,q + 5\,q^2 - 3\,q^3 + q^4$

```
(K ↦ Expand[J[K]] == Jones[K][q]) /@ AllKnots[{3, 10}]
```

Testing the Reidemeister Moves

```
SetAttributes[P, Orderless];
TJ[t2_] := Module[{t3, t4, t5, t6},
  t3 = t2 /. {
     Xp[i_, j_, k_, l_] :→ q * P[i, j] P[k, l] - q^2 * P[i, l] P[j, k],
     Xm[i_, j_, k_, l_] :→ -q^(-2) * P[i, j] P[k, l] + 1/q * P[i, l] P[j, k]
    };
  t4 = Expand[t3];
  t5 = t4 //. P[a_, b_] P[b_, c_] :→ P[a, c];
  t6 = t5 /. {
     P[i_, i_] :→ (q + 1/q),
     P[i_, j_]^2 :→ (q + 1/q)
    };
  Simplify[t6]
 ]
```




```
Xp[2, 2, 3, 1] // TJ
```

```
Xm[1, 2, 2, 3] // TJ
```




```
TJ /@ {Xp[1, 5, 2, 4] Xm[2, 5, 3, 6], Xp[4, 3, 5, 2] Xm[5, 1, 6, 2]}
```




```
TJ /@ {Xp[4, 2, 5, 1] Xp[7, 3, 8, 2] Xp[8, 6, 9, 5], Xp[7, 5, 8, 4] Xp[8, 2, 9, 1] Xp[5, 3, 6, 2]}
```

```
%[[1]] == %[[2]]
```

Analyzing a knot suggested by David Vincent




```
K1 = PD[X[6, 14, 7, 13], X[14, 6, 15, 5], X[4, 7, 5, 8], X[8, 15, 9, 16],
   X[16, 3, 1, 4], X[12, 9, 13, 10], X[10, 1, 11, 2], X[2, 11, 3, 12]];

J1 = J[K1]

Select[AllKnots[{3, 8}], J[#] == J1 &]
```